

Introduction:

Two types of files in QBASIC:

1. Program file :

Program file contains a set of instructions. (QBASIC Commands). Extension is ".bas"

2. Data file :

Data file is the file that stores data permanently which are required for data processing. Extension is ".dat" or any three letter extension. Data stored in a row, which is known as **record**. Data field may contain many records. Each Data of record is known as **field**.

Opening a data file:

OPEN "student.dat" FOR OUTPUT AS #1

Open in OUTPUT mode - new data file created and assigned 1 as a file number (File number can be 1 to 255).

OPEN "student.dat" FOR INPUT AS #2

Opens an existing data file - can retrieve records of data and manipulate info.

OPEN "student.dat" FOR APPEND AS #3

Opens an existing data file - can add more records to the end of the file.

Closing a data file:

CLOSE #1, #3

Closes both data files which have file numbers 1 and 3

CLOSE

Closes all opened data files.

Write, Reading, Append

Writing Data to a data file:

WRITE #1, Naam, Address, Age (Variables)

Write in a sequential data file

Reading Data to a data file:

INPUT #1, Naam\$, Address\$, Age

Retrieves data of the file number 1 in sequential order and stores them in variables.

File Modes

Update to a data file:

To change the value of fields or delete some records or insert some new records. The process of changing information of data file is known as updating. To update data or record of a sequential data file, follow these tasks.

1. Open the original file in INPUT mode to read data.
2. Open a new temporary file in OUTPUT mode to store necessary data.
3. Search the data of the original file & write the required data in the temp. file.
4. Repeat step "3" until all the data is written into the temporary file.
5. Close both files.
6. Delete the original file
7. Assign name of original file to the temporary file.
8. Close the file.

Example 1:

```
CLS
DIM Naam AS STRING
DIM Address AS STRING
DIM Age AS INTEGER
OPEN "info.dat" FOR OUTPUT AS #1
Naam = "Neha"
Address = "Harisiddhi"
Age = 14
WRITE #1, Naam, Address, Age
'Writing a data to new created file info.dat
CLOSE #1
END
```

Example 2:

```
REM open new file and input data until user wants
CLS
OPEN "info1.dat" FOR OUTPUT AS #1
'Create file name info1.dat
DO
Input "Enter name "; Naam$
Input "Enter address "; Address$
Input "Enter age "; Age
WRITE #1, Naam$, Address$, Age
Input "Do you want more(y/n)"; choice$
loop while Ucase$(choice$) = "Y"
CLOSE #1
END
```

Example 3

```

REM display data from data file (read)
CLS
OPEN "info1.dat" FOR INPUT AS #1
'Open file info1.dat
PRINT "Name", "Address", "Age"
DO WHILE NOT EOF (1)
INPUT #1, Naam$, Address$, Age
PRINT Naam$, Address$, Age
LOOP
CLOSE #1
END

```

Example 4:

```

REM display selected data from data file (read)
CLS
OPEN "info1.dat" FOR INPUT AS #1
'Open file info1.dat
PRINT "Name", "Address", "Age"
DO WHILE NOT EOF(1)
INPUT #1, Naam$, Address$, Age
IF UCASE$(Naam$) = "RAM" THEN
PRINT Naam$, Address$, Age
END IF
LOOP
CLOSE #1
END

```

Example 5:

```

REM retrieves records of a data file when you don't
know what the fields in the file are
CLS
OPEN "info1.dat" FOR INPUT AS #2
'Open file info1.dat
DO WHILE NOT EOF(2)
LINE INPUT #2, B$
'Retrieves all data of the file in sequential order
PRINT B$
LOOP
CLOSE #2
END

```

Example 6:

```

REM Adding data to a sequential file
CLS
OPEN "info1.dat" FOR APPEND AS #1
'Opens an existing data file to add more records to the
end of the file
DO
Input "Enter name "; Naam$
Input "Enter address "; Address$
Input "Enter age "; Age
WRITE #1, Naam$, Address$, Age
'Writing user input data to file
Input "Do you want more(y/n)"; choice$
Loop until Ucase$(choice$) = "N"
CLOSE #1

```

END

Example 7:

```

REM File Management Statements and Function
CLS
'FILES "*.dat"
'FILES "c:\*.*"
'KILL "l.dat"
'SHELL
END

```

Example 8:

```

REM modifying the contents of a data file
CLS
OPEN "INFO1.dat" FOR INPUT AS #1
'Open file info1.dat
OPEN "INFO2.dat" FOR OUTPUT AS #2
'Create file name info2.dat
DO WHILE NOT EOF(1)
INPUT #1, Naam$, Address$, Age
'Retrieves data of the file in sequential order
Age = Age + 20
WRITE #2, Naam$, Address$, Age
'Writing changed data to file
LOOP
CLOSE #1, #2
'KILL "info1.dat"
'NAME "Info2.dat" AS "info1.dat"
END

```

Example 9:

```

REM deleting the records
CLS
OPEN "INFO1.dat" FOR INPUT AS #1
'Open file info1.dat
OPEN "INFO2.dat" FOR OUTPUT AS #2
'Create file info1.dat

INPUT "Enter the Name to Delete: ", NaamDelete$
DO UNTIL EOF(1)
INPUT #1, Naam$, Address$, Age
'Retrieves data of the file in sequential order
IF Naam$ <> NaamDelete$ THEN
WRITE #2, Naam$, Address$, Age
'Writing found data to file
END IF
LOOP
CLOSE #1, #2
KILL "info1.dat"
NAME "Info2.dat" AS "info1.dat"
END

```

Example 10:

REM Search Data

```

CLS
OPEN "INFO1.dat" FOR INPUT AS #1
INPUT "Enter the Name to Search: ", NaamSearch$
DO UNTIL EOF(1)
INPUT #1, Naam$, Address$, Age
IF Naam$ = NaamSearch$ THEN
    PRINT Naam$, Address$, Age
    found = 1
    EXIT DO
END IF
LOOP
CLOSE #1
IF found = 0 THEN
    PRINT "A matching record is not found."
END IF
END

```

Project :

REM program of Telephone directory

REM list of variables used

'naam Name

'd District

't Tole

'p Phone Number

```

DECLARE SUB add () 'Record Adding module
DECLARE SUB dis () 'Record display module
DECLARE SUB ser () 'Record Search Module
DECLARE SUB edit () 'record editor module
DECLARE SUB del () 'record delete module

```

REM loading process

```

CLS
'SCREEN 12
'LOCATE 6, 35
PRINT "Loading..."
'COLOR 8
'LINE (38, 108)-(570, 132), 8, B
'LINE (70, 110)-(568, 130), 8, B
'A = 71
'FOR k = 1 TO 20
'    SLEEP 1
'    LINE (A, 111)-(A + 20, 129), 8, BF
'    A = A + 25
'NEXT k

```

```

CLS
PRINT "Date: "; DATE$
PRINT
PRINT "Telephone Directory"
PRINT

```

```

PRINT "MENU"
PRINT
PRINT "1. Add Records"
PRINT "2. Display Records"
PRINT "3. Search Records"
PRINT "4. Edit Records"
PRINT "5. Delete Records"
PRINT "6. Exit"
INPUT "Enter your choice (1/2/3/4/5/6)"; ch
SELECT CASE ch
    CASE 1
        CALL add
    CASE 2
        CALL dis
    CASE 3
        CALL ser
    CASE 4
        CALL edit
    CASE 5
        CALL del
    CASE 6
        PRINT
        PRINT "Thank you for using this program!!!"
END SELECT

SUB add
    PRINT
    PRINT "Adding..."
END SUB

SUB del
    PRINT
    PRINT "Deleding..."
END SUB

SUB dis
    PRINT
    PRINT "Displaying..."
END SUB

SUB edit
    PRINT
    PRINT "Editing..."
END SUB

SUB ser
    PRINT
    PRINT "Searching..."
END SUB

```